KfK 3909
März 1985

Matrix Correction for PIXE in Biomedical Samples

D. Heck, E. Rokita
Institut für Kernphysik

Kernforschungszentrum Karlsruhe

KERNFORSCHUNGSZENTRUM KARLSRUHE

Institut für Kernphysik

KfK 3909

Matrix Correction for PIXE in Biomedical Samples

D. Heck and E. Rokita[+]

Kernforschungszentrum Karlsruhe GmbH, Karlsruhe

[+] on leave from: Institute of Physics, Jagellonian University, Krakow, Poland

Kernforschungszentrum Karlsruhe GmbH
ISSN 0303-4003

Abstract

This report describes the programs MATRIX2, STPPWRF2 and MUROFIT, which are used for the calculation of the matrix correction factors, which must be applied to concentrations determined by PIXE (Proton induced X-ray emission). The correction takes into account the slowing down of the protons along their path through the specimen, which causes a decreasing X-ray production along this path. Moreover these X-rays are attenuated penetrating the specimen towards to the X-ray-detector. The matrix correction factors regard these effects in dependence on the proton impact energy, the specimen and detector geometry, the specimen composition and the energies of the interesting X-rays.

Matrix-Korrektur für PIXE in biologisch-medizinischen Proben

Zusammenfassung

Dieser Bericht beschreibt die Programme MATRIX2, STPPWRF2 und MUROFIT, die zur Berechnung der Matrixkorrekturfaktoren benutzt werden. Mit diesen Korrekturen müssen Konzentrationswerte aus PIXE-Messungen versehen werden (PIXE = proton induced X-ray emission). Die Korrekturen berücksichtigen das Abbremsen der Protonen entlang ihres Weges durch die Probe, was eine abnehmende Röntgenproduktion entlang dieses Weges verursacht. Außerdem werden diese Röntgenstrahlen beim Durchqueren der Probe zum Detektor abgeschwächt. Die Matrix-Korrekturfaktoren berücksichtigen diese Effekte in Abhängigkeit der Protonenenergie, der Geometrie von Probe und Detektor, der Probenzusammensetzung aus ihren Hauptbestandteilen und den Energien der interessierenden Röntgenstrahlen.

Introduction

Following refs. [1,2] we write for thin target PIXE experiments the detected x-ray yield Y(Z):

$$Y(Z) = \left(\frac{N_{av}\sigma_z(E_0)\omega_z b_z \varepsilon_z}{A_Z}\right) \cdot N \cdot M_a = K(Z) \cdot N \cdot M_a \qquad (1.1)$$

where N protons with energy E_0 hit onto a target of areal density M_a, with atomic number Z and atomic mass A_z; $\sigma_z(E_0)$ is the ionization cross section at energy E_0, ω_z the fluorescence yield, b_z the X-ray branching ratio, ε_z the detector efficiency and N_{av} Avogadro's number. The expression in brackets may be summarized to the so called [1] thin-target sensitivity factor K(Z). If the target has a finite tickness, by which the protons are slowed down and the X-rays partially absorbed on their way to the detector, and the element Z is abundant only in the weight concentration C_z, eq. (1.1) is replaced by integrating over the proton path:

$$Y(Z) = \left(\frac{N_{av} \cdot \omega_z \cdot b_z \cdot \varepsilon_z}{A_Z}\right) \quad N \cdot C_Z \int_{E_0}^{E_1} \frac{\sigma_z(E)\ T_Z(E)\,dE}{S(E)} \qquad (1.2)$$

Here $T_z(E)$ is the X-ray attenuation and S(E) the matrix stopping power. Introducing again the thin target sensitivity factor K(Z), we arrive at

$$Y(Z) = K(Z) \cdot N \cdot C_z \cdot t \cdot \frac{\int_{E_0}^{E_1} (\sigma_z(E) \cdot T_z(E)/S(E))\,dE}{t \cdot \sigma_z(E_0)} \qquad (1.3)$$

The integral extends over the thickness t of the sample and the protons will be slowed down to E_1 when leaving the target. With the geometry of the Karlsruhe nuclear microprobe [3] we restrict the target angle α to 0° (perpendicular to the beam, see ref.[1]) and the detector angle θ to 45° backwards. Then the X-ray transmission $T_Z(E)$ becomes

$$T_Z(E) = \exp\left\{-\left(\frac{\mu}{\rho}\right)_Z \cdot \frac{1}{\cos\theta} \cdot \int_{E_0}^{E_1} \frac{dE}{S(E)}\right\} \qquad (1.4)$$

For a real matrix with i constituents of mass fractions C_i, the attenuation coefficient $(\mu/\rho)_z$ must be replaced by $\sum_i C_i (\mu/\rho)_{iz}$ and similar $S(E)$ by $\sum_i C_i S_i(E)$.

We define the matrix correction factor MCF by

$$MCF = \frac{t\ \sigma_z(E_0)}{\int_{E_0}^{E_1} (\sigma_z(E) \cdot T_z(E)/S(E))\,dE} \qquad (1.5)$$

which gives the ratio of the X-ray number detected from an idealized target to the X-ray number recorded from a real target of the same thickness. The factor MCF must be calculated for the proton impact energy E_0, the target thickness t, the matrix composition C_i and the trace element atomic number Z. This is done in computer programs, using tabulated values of the mass attenuation coefficients $(\mu/\rho)_i$ [4], the stopping powers $S_i(E)$ [5] and an empirical expression for the ionization cross section $\sigma_Z(E)$ [6]. The results show, that the factors MCF increase linearly for organic targets up to $\sim$ 2 mg/cm², so we may approximate it by

$$MCF = 1 + (\rho \cdot t)\ s_Z \qquad (1.6)$$

with s_Z the element dependent slope [7] of the matrix correction factor.

2. The program MATRIX2

This program is written for the evaluation of matrix correction factors (MCF). It is organized in the main program (performing in- and output) and the subroutines WSPF (calculation of MCF), RANGE (proton energy as function of range), SIGMA (ionization cross section), ATTEN (X-ray attenuation), and ARSIMP (integration using Simpson's rule). A complete listing is given in Appendix 1.

2.1 Input instructions (card images)

1)	TEXT	(15A4)	60 characters heading line
2)	IPAR	(I1)	Steering parameter: = 0 MCF for given target thickness = 1 Table MCF for 20 target thicknesses = 2 like 1, additionally relative MCF = 3 ionization cross sections
3)	EPP	(F10.0)	Proton impact energy in MeV. Max. value 4 MeV.
4)*	SF	(F10.0)	Thickness step in mg/cm²
5)*	$W_1 \ldots W_{10}$	(5F10.0)	Matrix composition; mass fractions of H,C,N,O,Na,P,S,Cl,K,Ca.
6)	N	(I10)	Number of elements, for which the MCF is calculated, ($N \leq 99$)
7)	N times:		
a]	for IPAR = 1,2,3		
	I_N, K_N	(2I10)	I = X-ray line type: 1 = K_α line, 2 = K_β line, 3 = L_α line K = atomic number of element $11 \leq K \leq 83$
b]	for IPAR = 0		
	I_N,K_N, T	(2I10, F10.0)	similar to a] with T = target thickness
8)**	IZ	(I10)	IZ = atomic number of reference element for relative MCF

---

* To be omitted for IPAR = 3

** Only for IPAR = 2

2.2 The subroutines

2.2.1 RANGE(EPP,DX1,Z)

It calculates the energy of the proton (impact energy EPP(MeV)) along its path up to the range Z (mg/cm²). The step length is DX1 (mg/cm²). The resulting energy values are stored in the COMMON array E(8000). The stopping power SP is calculated by the formula

$$SP = a \cdot (E)^{b} \exp(-c \cdot E) + c_2 \cdot E + d \qquad (2.1)$$

The parameters a,b,c,c_2 and d are fitted for the various matrix elements of vector W by the program STPPWRF2 (see sect. 3).

2.2.2 SIGMA(I,K,EP,S)

It calculates the X-ray ionization cross section S as function of the proton energy EP for the element with atomic number K and the X-ray line type I(1 = K-lines, 2 = L-lines), using the polynominal expression of ref. [6].

2.2.3 ATTEN(EX,A)

It calculates the X-ray mass attenuation coefficient A as function of the X-ray energy EX. The interpolation formula

$$A = x_2 \cdot EX^{-x_1} \cdot conv \qquad (2.2)$$

is used (conv = conversion factor from barn/atom to cm²/g). The parameters x_1 and x_2 are fitted for the matrix elements by the program MUROFIT (see sect. 4) below and above the K-absorption edge.

2.2.4 ARSIMP(J,DP,A,Z)

It calculates the integral $Z = \int A\, dp$ using Simpson's rule over J intervals of width DP.

3. The program STPPWRF2

3.1 General considerations

This program is designed to fit the coefficients of the approximation formula eq. (2.1) of sect. 2.2.1 to the tabulated values of ref. [5]. The choice of this formula is intuitive. Tests showed that

$$\lim_{E \to 0} \quad SP > 0$$

guarantees a good convergence of the range of the protons. If different formulae are used, they have to fulfil at minimum

$$\lim_{E \to 0} \quad \frac{\partial SP}{\partial E} > 0 \ ,$$

otherwise the protons penetrate with infinite small energies very large distances through bulk matter without being stopped completely. The formula (2.1) represents all stopping powers of interest sufficiently well in the energy range 4 MeV down to 100 keV. Only below 100 keV the deviations from the tabulated values are considerably. But this is easily tolerated, as in this energy range the contribution to the integral of eq. (1.5) is extremely small caused by the steep slope of the cross section σ, which decreases from 1 MeV to 100 keV by two orders of magnitude, for $E_x > 3$ keV even more than 3 orders of magnitude.

As the energy intervals of the tabulated values of ref. [5] follow coarsely a logarithmic scale, a fit with equal weighting of all values would overemphasize the low energy tail of the curve and represent the important range between 1 and 4 MeV rather bad. Therefore the tabulated values have been furnished with a weighting function $W \sim E^2$ to improve the precision of the formula eq. (2.1) for energies above 1 MeV. A critical test on the quality of the approximation of eq. (2.1) and the fitted coefficients is the proton range. The ranges of 3.225 MeV protons calculated by the subroutine RANGE (sect.3.2.1) deviate considerably less than 1 % from the tabulated ranges [5] for the materials H,C,N,O,Ca, H_2O, and $(CH_2)_n$.

3.2 Subroutines

The program STPPWRF2 makes use of the fitting routine SQUFT5, which is an improved version of SQUFIT [8] and stored on the LOAD.IAK program library of the IBM computer of KfK. In the subroutine SPFIT the function value FF and the derivatives of eq.(2.1) are calculated:

$$\frac{\partial SP}{\partial a} = A(1) = E^b \cdot \exp(-cE)$$

$$\frac{\partial SP}{\partial b} = A(2) = a \cdot E^b \exp(-cE) \cdot \ln E$$

$$\frac{\partial SP}{\partial c} = A(3) = -aE \cdot E^b \exp(-cE)$$

$$\frac{\partial SP}{\partial d} = A(4) = 1$$

$$\frac{\partial SP}{\partial c_2} = A(5) = E$$

The subroutines LESCOP, LESINT and LESEXP perform a FORMAT-free input reading and are also stored on the LOAD.IAK library. A complete listing of STPPWRF2 is given in Appendix 2.

3.3 Input instructions (FORMAT-free)

1)	TEXT	(15A4)	60 characters heading line
2)	IMAX		Number of tabulated values to be fitted
3)	IMAX times:		
	E_i, SP_i		Couples of energy (keV) and stopping power [MeV/(mg/cm²)]. They should be arranged with increasing energy.
4)	a,b,c,d,c_2		Starting values for fit.

4. The program MUROFIT

4.1 General considerations

This program is designed to fit the coefficients, which are used to approximate the X-ray attenuation cross sections A of ref. [4] as function of the X-ray energy E_x by the expression

$$A = x_2 \cdot E_x^{-x_1}$$

The cross sections A are related to the mass attenuation coefficients μ/ρ by the conversion factor conv [4]

$$\mu/\rho = A \cdot \text{conv}$$

By the occurence of an absorption edge it is necessary to fit sets of coefficients below and above the edge energy. Similar to sect.3.1 the tabulated values get weights, here we use the weighthing function

$$W \sim 1/\sqrt{A}$$

which has the effect, that high A values are fitted with a relative smaller error $\Delta A/A$ than lower ones. Regarding fig. 5 of ref. [1] the effect of fractional changes of μ/ρ onto the calculated X-ray yield and hence the matrix correction factors MCF is largest for high μ/ρ values. For $\mu/\rho \gtrsim 300$ the fractional change of the MCF is almost equal to the fractional change in μ/ρ . For low μ/ρ values $\lesssim 10$ the fractional changes of the MCF's are less than 0.1 the fractional changes of μ/ρ. The examination of the fitted A values in the range 1 keV $\leq E_x \leq$ 20 keV shows that the influence of the uncertainties of A causes errors in the MCF generally well below 5 %. Only for Oxygen at $\sim$ 4 keV this border is reached and exceeded for Na between 3 and 10 keV.

4.2 Subroutines for MUROFIT

The program MUROFIT makes use of the fitting routine SQUFT5, as described in sect. 3.2. The derivatives calculated in the subroutine MRFIT are

$$\frac{\partial A}{\partial x_1} = A(1) = -x_2 \cdot E_x^{-x_1} \cdot \ln(E_x)$$

$$\frac{\partial A}{\partial x_2} = A(2) = E_x^{-x_1}.$$

A complete listing of MUROFIT is given in Appendix 3

4.3 Input instructions (FORMAT-free)

1)	TEXT	(15A4)	60 characters heading line
2)	IMAX		Number of tabulated values to be fitted
3)	IMAX times:		
	E_i, A_i		Couples of energy (keV) and attenuation cross sections A (barns/atoms). They should be arranged with increasing energy.
4)	x_1, x_2		Starting values for fit.

Literature:

[1] J.L. Campbell, J.A. Cookson, H. Paul, Nucl. Instr. Meth. 212 (1983) 427

[2] B. van Oystaeyen, G. Demortier, Nucl. Instr. Meth. 215 (1983) 299

[3] D. Heck, KfK-Nachrichten 16 (1984) 145;
D. Heck, Atomkernenergie - Kerntechnik 43 (1985) (in print)

[4] WM.J. Veigle, Atomic Data Tables 5 (1973) 51

[5] L.C. Northcliffe, R.F. Schilling, Nucl. Data Tables A7 (1970) 233

[6] S.A.E. Johansson, T.B. Johansson, Nucl. Instr. Meth. 137 (1976) 473

[7] D. Heck, E. Rokita, Nucl. Instr. Meth. B3 (1984) 259

[8] D. Braess, unpublished (1965)

APPENDIX 1

```
C***************  MATRIX2   ******************************************
C  FORTRAN PROGRAM FOR CALCULATION OF MATRIX CORRECTION FACTORS (MCF)
C  AND X-RAY CROSS SECTION (CS) FOR PROTON (PROTON ENERGY EPP UP TO 4
C  MEV)INDUCED X-RAYS IN A TARGET COMPOSED OF H,C,N,O,NA,P,S,CL,K,CA
C  (TARGET COMPOSITION GIVEN IN VECTOR W)    VERSION 2   JAN 85/HECK
C---------------------------------------------------------------------
      DIMENSION T1(20),F1(20),F2(99,99),IF3(2,99),TYP(3),WA(10),A(10)
      COMMON  W(10),ALFA,BETA,DEL
      COMMON E(8000)
      DIMENSION TEXT(15)
      CHARACTER*2 ELEMNT(83)
      DATA    ELEMNT/'H ','HE','LI','BE','B ','C ','N ','O ','F ',
     *'NE','NA','MG','AL','SI','P ','S ','CL','AR','K ','CA','SC',
     *'TI','V ','CR','MN','FE','CO','NI','CU','ZN','GA','GE','AS',
     *'SE','BR','KR','RB','SR','Y ','ZR','NB','MO','TC','RU','RH',
     *'PD','AG','CD','IN','SN','SB','TE','J ','XE','CS','BA','LA',
     *'CE','PR','ND','PM','SM','EU','GD','TB','DY','HO','ER','TM',
     *'YB','LU','HF','TA','W ','RE','OS','IR','PT','AU','HG','TL',
     *'PB','BI'/
      DATA ALFA/0.7853982/
C     DATA BETA/0.7853982/
      DATA BETA/0.0000000/
      DATA DEL/0.01/,TYP/' KA ',' KB ',' LA '/
      DATA A/1.008,12.011,14.007,15.999,22.990,
     *  30.974,32.064,35.453,39.102,40.08/
      READ(5,101)TEXT
  101 FORMAT(15A4)
      READ(5,601) IPAR
 601  FORMAT(I1)
C******IPAR = 0  MCF FOR GIVEN THICKNESS
C            1  TABLE OF MCF VALUES FOR GIVEN EPP AND TARGET COMPOSITION
C            2  1 AND RELATIVE MCF
C            3  IONISATION CROSS SECTION CALCULATION
      READ (5,400) EPP
C******PROTON ENERGY(F10.0)
      IF(IPAR.EQ.3) GO TO 444
      READ(5,400) SF
C******SF ONLY FOR 1 AND 2 (FORMATF10.0) THICKNESS STEP (MG/CM**2)
      WRITE(6,555)
  555 FORMAT(1H1)
      IF(EPP.GT.4.03)EPP=4.0
      WRITE(6,200) EPP
 200  FORMAT(1X,27H      PROTON  ENERGY        ,F6.3,2X,3HMEV/)
      BE=ALFA*57.29578
      WRITE(6,300) BE
 300  FORMAT(1X,21H      DETECTOR ANGLE ,F10.2/)
      BE1=BETA*57.29578
      WRITE(6,301) BE1
 301  FORMAT(1X,21H      TARGET ANGLE   ,F10.2/)
      WRITE(6,102)TEXT
  102 FORMAT(1H ,6HTARGET,2X,15A4)
```

```
      READ(5,400) W
 400  FORMAT(5F10.0)
C****TARGET COMPOSITION (5F10.0) MASS FRACTION OF H,C,N,O,NA,P,S,CL,K,CA
      SUMW=0.
      DO 114 L=1,10
  114 SUMW=SUMW+W(L)/A(L)
      DO 112 L=1,10
  112 WA(L)= 100.*(W(L)/A(L))/SUMW
      WRITE(6,500) (( W(I),WA(I)),I=1,10)
 500  FORMAT(1X, '  TARGET   COMPOSITION',/1X,'ELEMENT   MASS FRACTION',
     *'   ATOMIC PERCENT',/1X,6H   H  ,2 F17.5,/1X,6H   C  ,2 F17.5,/1X,
     *   6H   N  ,2 F17.5,/1X,6H   O  ,2 F17.5,/1X,6H   NA ,2 F17.5,/1X,
     *   6H   P  ,2 F17.5,/1X,6H   S  ,2 F17.5,/1X,6H   CL ,2 F17.5,/1X,
     *   6H   K  ,2 F17.5,/1X,6H   CA ,2 F17.5,//)
      READ(5,700) N
C*****NUMBER OF ELEMENTS (FORMATI10)
      IF(N.GT.99)N=99
      DP=DEL/COS(BETA)
      CALL RANGE(EPP,DP,ZP)
      WRITE(6,602) ZP
 602  FORMAT(1X,19H  PROTON RANGE  ZP=,F10.2,7H MG/CM2,/)
      IF(IPAR.EQ.0) GO TO 111
      WRITE(6,557)
 557  FORMAT(1X,30X,27HMATRIX  CORRECTION  FACTORS,/,1X,29X,29(1H=),////
     *,1X,3X,17HZ = ATOMIC NUMBER,  10X,  'THICKNESS - MG/CM**2',//)
      DO 999 L=1,N
      READ(5,700) I1,K1
 700  FORMAT(2I10)
C*****I1 = 1 - K ALFA, =2 - K BETA, =3 L ALFA
C     K1 ATOMIC NUMBER (FORMAT 2I10) K1>=11
      IF(K1.LE.10)K1=11
      IF(K1.GT.83)K1=83
      IF(I1.GT.3)I1=3
      IF(I1.LT.1)I1=1
      IF3(1,L)=K1
      IZ1=K1
      L2=I1
      IF3(2,L)=L2
      DO 888 M=1,20
      T1(M)=M*SF
      CALL WSPF(I1,K1,T1(M),ZP,F)
      F1(M)=F
      IF(F1(M).LT.1.0) F1(M)=1.0000
 888  CONTINUE
      DO 777 I=1,20
      RZ=ZP-T1(I)
      IF(RZ.LT.0.0) GO TO 778
 777  CONTINUE
 778  N15=I+1
      DO 779 J=N15,20
 779  F1(J)=0.00000
      DO 560 I=1,20
 560  F2(L,I)=F1(I)
      IF(L.EQ.1) WRITE(6,901) (T1(IJ),IJ=1,20)
      WRITE(6,902)ELEMNT(IZ1),IZ1,TYP(L2),(F1(IJ),IJ=1,20)
 999  CONTINUE
```

```
      IF(IPAR.EQ.1) GO TO 333
C_____CALCULATION OF RELATIVE INTENSITIES
      READ(5,700)IZ5
C*****ATOMIC NUMBER FOR CALCULATION RELATIVE INTENSITIES FORMAT(F10.0)
      WRITE(6,564)
 564  FORMAT(1X,///,1X,//////,1X,30X,37HRELATIVE  MATRIX  CORRECTION  FA
     *CTORS,/,1X,29X,39(1H=),////)
      DO 561 J=1,N
      IF(IZ5.EQ.IF3(1,J))  N5=J
 561  CONTINUE
      DO 565 J=1,20
      IF(F2(N5,J).EQ.0.0) F2(N5,J)=1.0
 565  CONTINUE
      WRITE(6,901) (T1(IJ),IJ=1,20)
 901  FORMAT(4X,3H Z ,3X,20(F5.2,1X),/1X,129(1H=),/)
      DO 562 L=1,N
      DO 563 J=1,20
 563  F1(J)=F2(L,J)/F2(N5,J)
      I5=IF3(2,L)
      I6=IF3(1,L)
      WRITE(6,902)ELEMNT(I6),IF3(1,L),TYP(I5),(F1(IJ),IJ=1,20)
 902  FORMAT(1X,A2,1X,I2,A4    ,20(F6.3),/)
 562  CONTINUE
      GO TO 333
C_____CALCULATION OF CORRECTION FOR GIVEN TARGET THICKNESS
 111  WRITE(6,800)
 800  FORMAT(1X,14HTARGET  THICK.,1X,14HATOMIC NUMBER ,6X,5HLINE ,6X,
     *12HMATRIX CORR./)
      DO 222 L=1,N
      READ(5,701) I1,K1,T1(L)
 701  FORMAT(2I10,F10.0)
C******I1,K1 AS ABOVE, T1 TARGET THICKNESS (FORMAT 2I10,F10.0)
      L1=I1
      CALL WSPF (I1,K1,T1(L),ZP,F)
      IF(F.LT.1.0) F=1.00000
      WRITE(6,900) T1(1),ELEMNT(K1),K1,TYP(L1),F
 900  FORMAT(1X,F14.5,1X,4X,A2,I10,10X,A4,6X,F14.5)
 222  CONTINUE
      GO TO 333
C_____CALCULATION OF X-RAY CROSS SECTIONS
 444  READ(5,700) N
C********
      DO 445 L=1,N
      READ(5,700) I1,K1
C********
      IZ3=K1
      L3=I1
      IF(I1.EQ.2) I1=1
      IF(I1.EQ.3) I1=2
      EP1=EPP*0.1
      DO 446 J=1,10
      T1(J)=EP1*J
      CALL SIGMA(I1,K1,T1(J),S)
 446  F1(J)=S
      IF(L.EQ.1)WRITE(6,447) (T1(I),I=1,10)
```

```
447  FORMAT(1X,/////,1X,30X,28H X-RAY CROSS SECTION (BARN) ,/,
    *30X,30(1H=),////,1X,'ELEMENT     PROTON ENERGY ',/,
    *4X,3H Z  ,3X,10(F6.3,6X),/,1X,129(1H=),/)
     WRITE(6,448)ELEMNT(IZ3),IZ3,TYP(L3),(F1(I),I=1,10)
448  FORMAT(1X,A2,1X,I2,A4   ,10(F12.6),/)
445  CONTINUE
333  STOP
     END
```

```
      SUBROUTINE WSPF(I,K,T,RP,FF)
C-----------------------------------------------------------------------
C     CALCULATES INTEGRAL OF PRODUCTION * ATTENUATION
C-----------------------------------------------------------------------
      REAL*4  ENX(2,73),A(8000),MU,DX,SIG
      COMMON W(10),ALFA,BETA,DEL
      COMMON E(8000)
C_____ENX= X-RAY ENERGIES OF K-ALPHA1, K-ALPHA 2 ETC LINES
C     UP TO Z=36,THAN K-ALPHA 1 AND L-ALPHA1 LINES,
C     UP TO Z=58, THAN ONLY L-ALPHA 1 LINES
C     FOR 11<= Z <=83
      DATA ENX /1.041,1.067 ,1.254,1.297, 1.487,1.553, 1.740,1.832,
     *          2.015,2.140,2.308,2.460,2.622,2.820,2.957,3.190,3.313,
     *3.589,3.691,4.012,4.090,4.460,4.510,4.931,4.952,5.427,5.414,5.946,
     *5.898,6.490,6.403,7.057,6.930,7.650,7.477,8.260,8.047,8.904,8.638,
     *9.620,9.251,10.31,9.885,11.05,10.543,11.76,11.221,12.58,11.923,
     *13.290,12.648,1.587,13.394,1.694,14.164,1.806,14.957,1.922,
     *15.774,2.042,16.614,2.166,17.478,2.293,18.410,2.424,19.278,2.558,
     *20.214,2.696,21.175,2.838,22.162,2.984,23.172,3.133,24.207,3.287,
     *25.270,3.444,26.357,3.605,27.471,3.769,28.610,3.937,29.802,4.111,
     *30.970,4.286,32.191,4.467,33.440,4.651,34.717,4.840,0.0,5.034,0.0,
     *5.230,0.0,5.431,0.0,5.636,0.0,5.846,0.0,6.059,0.0,6.275,0.0,
     *6.495,0.0,6.720,0.0,6.948,0.0,7.181,0.0,7.414,0.0,7.654,0.0,
     *7.898,0.0,8.145,0.0,8.396,0.0,8.651,0.0,8.910,0.0,9.173,0.0,9.441,
     *0.0,9.711,0.0,9.9870,0.0,10.266,0.0,10.549,0.0,10.836/
      DP=DEL/COS(BETA)
      DX=DEL/COS(ALFA)
      KK=K-10
      IF(I.EQ.3) I=2
      EX=ENX(I,KK)
C_____ ATTENUATION CALCULATION
      CALL ATTEN(EX,MU)
      G=T
      IF(G.GT.RP) G=RP-DP
      J=1+INT(G/DP+0.5)
      DO 22 II=1,J
      CALL SIGMA(I,K,E(II),SIG)
   22 A(II)=SIG* EXP(-MU*(II-1)*DX)
C_____INTEGRATION
      CALL ARSIMP(J,DP,A,R)
      IF(R.LT.0.001) R=100.0
      FF=G*A(1)/R
      RETURN
      END
```

```
      SUBROUTINE ATTEN(EX,A)
C-----------------------------------------------------------------------
C     CALCULATES MASS ATTENUATION COEFFICIENT
C-----------------------------------------------------------------------
      REAL*4   CONV(10),X(2,20),A,Y,EDGE(20)
      COMMON W(10),ALFA,BETA,DEL
      DATA CONV/0.5975,.0514,.04300,0.03764,0.02620,
     *          0.01944,0.01878,0.01699,0.01540,0.01503/
      DATA X/-1.912022,7.3553125,-0.2176986,1.0612276,
     *      -2.914241 ,44554.46 ,-1.        ,0. ,
     *      -2.861572 ,80630.94 ,-1.        ,0. ,
     *      -2.770826 ,126310.8 ,-1.        ,0. ,
     *      -2.581572 ,24985.34 ,-2.681613 ,379684.9,
     *      -2.664633 ,99390.31 ,-2.700791 ,1151963.,
     *      -2.666108 ,130984.7 ,-2.714993 ,1476708.,
     *      -2.648788 ,168982.7 ,-2.710775 ,1822838.,
     *      -2.651941 ,269983.2 ,-2.725275 ,2779649.,
     *      -2.662639 ,335939.5 ,-2.720058 ,3304279./
      DATA EDGE/3.0,100.,100.,100.,1.072,2.142,2.470,2.822,3.607,4.038/
      A=0.
      DO 10 I=1,10
      IF(EX.LE.EDGE(I)) J=1
      IF(EX.GT.EDGE(I)) J=0
      Y=W(I)*X(2,2*I-J)*EX**X(1,2*I-J)*CONV(I)
   10 A=A+Y
      A=A*0.001
      RETURN
      END
```

```
      SUBROUTINE RANGE(EP,DX1,Z)
C-----------------------------------------------------------------------
C     CALCULATES PROTON RANG AND STOPPING POWERS
C-----------------------------------------------------------------------
      COMMON W(10),ALFA,BETA,DEL
      COMMON E(8000)
      DIMENSION A(10),B(10),C(10),D(10),C2(10)
      DATA A/6.191698   ,0.8990399 ,2.035140  ,1.783269  ,0.4354221,
     *          0.3237287 ,0.3159022 ,0.2980154 ,0.2747611 ,0.2724180/
      DATA B/0.6171969  ,0.2028544 ,0.7121236 ,0.6666598 ,0.04988067,
     *          0.02253590,0.03574158,0.03356688,0.04033075,0.05630720/
      DATA C/2.536004   ,2.174846  ,3.114668  ,3.014448  ,1.908115,
     *          1.830215  ,1.867549  ,1.835310  ,1.798602  ,1.858340 /
      DATA D/0.5596605  ,0.1751211 ,0.1816534 ,0.1704270 ,0.1432685,
     *          0.1308460 ,0.1297293 ,0.1267248 ,0.1206127 ,0.1197922/
      DATA C2/0.08481431,0.02289066,0.02506582,0.02298433,0.01841684,
     *          0.01653278,0.01648522,0.01600078,0.01483174,0.01488819/
      J=1
      E(1)=EP
  9   IF(E(J).LE.0.0)GO TO 60
      SP=0.
      DO 10 I=1,10
 10   SP=SP+W(I)*(A(I)*(E(J)**B(I))*EXP(-C(I)*E(J))+D(I)-C2(I)*E(J))
      J=J+1
      IF(J.GT.8000) GO TO 61
      DE=SP*DX1
      E(J)=E(J-1)-DE
      GO TO 9
   61 WRITE(6,62)E(J)
   62 FORMAT(1H ,'STORAGE E(8000) EXCEEDED AT PROTON ENERGY =',F10.5)
 60   E(J)=0.0
      J=J-1
      Z=J*DX1
      RETURN
      END
```

```
      SUBROUTINE SIGMA(I1,K1,EP,S)
C-----------------------------------------------------------------------
C     CALCULATES SIGMA (IONIZATION CROSS SECTION)
C     AFTER JOHANSSON +JOHANSSON
C-----------------------------------------------------------------------
      REAL*4    WSPB(6,2),EL(2,73),Y,W,S,U,X,YP,Z,Z1
C*****   WSPB:POLYNOMIAL COEFFICIENTS FOR CROSS SECTION
      DATA WSPB/2.0471,-0.0065906,-0.47448,0.09919,0.046063,0.0060853,
     *       3.6082,0.37123,-0.36971,-0.000078593,0.0025063,0.0012613/
C*****   EL: IONIZATION ENERGIES SODIUM TO BISMUTH
      DATA EL/1.08,0.034, 1.303,0.049, 1.559,0.072, 1.838,0.098,
     *        2.142,0.128,2.470,0.163,2.819,0.202,3.203,0.245,3.607,
     *0.294,4.038,0.349,4.496,0.406,4.964,0.454,5.463,0.512,5.988,0.574,
     *6.527,0.639,7.111,0.708,7.709,0.779,8.331,0.853,8.980,0.933,9.660,
     *1.022,10.368,1.117,11.103,1.217,11.868,1.323,12.652,1.434,13.475,
     *1.552,14.323,1.675,15.201,1.806,16.106,1.941,17.037,2.079,17.998,
     *2.220,18.987,2.374,20.002,2.523,21.054,2.677,22.118,2.837,23.224,
     *3.002,24.347,3.172,25.517,3.352,26.712,3.538,27.928,3.729,29.190,
     *3.928,30.486,4.132,31.809,4.341,33.164,4.559,34.579,4.782,35.959,
     *5.011,37.410,5.247,38.931,5.489,40.449,5.729,41.998,5.968,43.571,
     *6.215,45.207,6.466,46.846,6.721,48.515,6.983,50.229,7.252,51.998,
     *7.519,53.789,7.850,55.615,8.074,57.088,8.364,59.969,8.652,61.303,
     *8.943,63.304,9.241,65.313,9.556,67.400,9.876,70.508,10.198,71.662,
     *10.531,73.860,10.869,76.097,11.211,78.379,11.559,80.713,11.919,
     *83.106,12.285,85.517,12.657,88.001,13.044,90.521,13.424/
      N=K1-10
      U=EP*1.0E3/1836.12/EL(I1,N)
      X=ALOG(U)
      Y=0.0
      DO 10 K=1,6
      YP=WSPB(K,I1)*X**(K-1)
  10  Y=Y+YP
      Z=ABS(Y)
      Z1=Y/Z
      IF(Z.LT.0.00000001) Z=0.00000001
      Y=Z1*Z
      W=EXP(Y)
C**      S: CROSS SECTION IN BARNS
      S=W/EL(I1,N)/EL(I1,N)*10000.
      RETURN
      END
```

```
      SUBROUTINE ARSIMP(J,DP,A,Z)
C-------------------------------------------------------------------------
C     CALCULATES INTEGRAL BY SIMPSON RULE
C-------------------------------------------------------------------------
      REAL*4    A(8000)
      Z= (A(1)+A(J))*0.5*DP
      IF(J-2) 103,102,104
  104 K5=J-2
      DO 101 I=1,K5
      R=DP*(A(I)+A(I+2)+4.*A(I+1))/6.0
  101 Z=Z+R
  102 RETURN
  103 Z=0.5*Z
      GO TO 102
      END
```

APPENDIX 2

```
C***************  STPPWRF2  ********************************************
C----- FITS STOPPING POWER ---VERSION 2-----------------------------
C      BY D.HECK,JAN 1985
C-------------------------------------------------------------------
      DIMENSION IDUMMY(1011)
      DIMENSION ITEXT(15)
      COMMON IMAX,Y(300),P(300),NX,NXG,X(32),FF,A(32),NB,NBG,NF,Q,PF,
     1FV(32),FFV(5),NT,PAR(300)
      EQUIVALENCE(IDUMMY(1),IMAX)
      EXTERNAL SPFIT
      DO 3  I=1,1011
    3 IDUMMY(I)=0
      WRITE(6,5)
    5 FORMAT(1H1)
      CALL LESCOP
C     NX=3
      NX=5
      NF=0
      NB=0
      NXG=0
      READ(5,100)ITEXT
  100 FORMAT(15A4)
      CALL LESINT(IMAX)
      DO 1 I=1,IMAX
C ALL ENERGIES IN KEV
      CALL LESEXP(PAR(I),Y(I))
      IF(Y(I).EQ.0.) Y(I)=0.00000001
      P(I)=PAR(I)**2
    1 CONTINUE
      CALL LESEXP(X(1),X(2),X(3),X(4),X(5))
      CALL SQUFT5(1,0,0,SPFIT,SPFIT,SPFIT)
      WRITE(6,112)
  112 FORMAT (1H1,'FIT 3A')
      WRITE(6,101)ITEXT
  101 FORMAT(1H ,15A4)
      WRITE(6,104) PF
  104 FORMAT(1H ,7HCHI**2 ,G15.7)
      WRITE(6,2) X(1),X(2),X(3),X(4),X(5)
    2 FORMAT(1H ,'COEFFICIENTS '/' A =',G14.7,3X,'B =',G14.7,
     *'C =',G14.7,'D =',G14.7,'C2 =',G14.7)
      WRITE(6,105)
  105 FORMAT(1H ,'ENERGY (KEV)',5X,'STOP-PWR',4X,'STOP-PWR FITTED',
     *3X,'DEVIATION %',1X)
      DO 6  I=1,IMAX
      CALL SPFIT(I)
      Y(I+100)=FF
    6 Y(I+200)=(Y(I+100)/Y(I)-1.)*100.
      WRITE(6,4)(PAR(I),Y(I),Y(I+100),Y(I+200),I=1,IMAX)
    4 FORMAT(1H ,4G15.7)
      STOP
      END
```

```
      SUBROUTINE SPFIT(I)
C-----------------------------------------------------------------------
C     CALCULATES FUNCTION FF AND DERIVATIVES A(I) FOR PROGRAM MUROFIT
C-----------------------------------------------------------------------
      COMMON IMAX,Y(300),P(300),NX,NXG,X(32),FF,A(32),NB,NGB,NF,Q,PF,
     *FV(32),FFV(5),NT,PAR(300)
      A(1) =(PAR(I)**X(2))*EXP(-X(3)*PAR(I))
      A(2) = X(1)*A(1)*ALOG(PAR(I))
      A(3) = -PAR(I)*A(1)*X(1)
      A(4) = 1
      A(5) = PAR(I)
      FF = A(1)*X(1)+X(4)+X(5)*PAR(I)
      RETURN
      END
```

APPENDIX 3

```
C*************** MUROFIT  ***********************************************
C----- FITS X-RAY ATTENUATION CROSS SECTIONS (BARNS/ATOM) ------------
C       BY D.HECK,JAN 1985
C---------------------------------------------------------------------
      DIMENSION IDUMMY(1011)
      DIMENSION ITEXT(15)
      COMMON IMAX,Y(300),P(300),NX,NXG,X(32),FF,A(32),NB,NBG,NF,Q,PF,
     1FV(32),FFV(5),NT,PAR(300)
      EQUIVALENCE(IDUMMY(1),IMAX)
      EXTERNAL MRFIT
      DO 3  I=1,1011
    3 IDUMMY(I)=0
      CALL LESCOP
      NX=2
      NF=0
      NB=0
      NXG=0
      READ(5,100)ITEXT
  100 FORMAT(15A4)
      CALL LESINT(IMAX)
      DO 1 I=1,IMAX
C ALL ENERGIES IN KEV
      CALL LESEXP(PAR(I),Y(I))
C******* ENERGY = PAR(I) IN KEV, ATTEN. CROSS SECT. Y(I) IN BARNS/ATOM
      IF(Y(I).EQ.0.) Y(I)=0.00000001
C---    P = WEIGHTING FACTOR
    1 P(I)=1./SQRT(Y(I))
      CALL LESEXP(X(1),X(2))
C******* STARTING VALUES
      CALL SQUFT5(1,0,0,MRFIT,MRFIT,MRFIT)
      WRITE(6,101)ITEXT
  101 FORMAT(1H1,15A4)
      WRITE(6,104) PF
  104 FORMAT(1H ,7HCHI**2 ,G15.7)
      WRITE(6,2) X(1),X(2)
    2 FORMAT(1H ,'COEFFICIENTS '/' X1 = ',G15.7,3X,'X2 = ',G15.7//)
      WRITE(6,105)
  105 FORMAT(1H ,'ENERGY (KEV)',6X,'MU/RO',8X,'MU/RO FITTED',4X,
     *'DEVIATION %')
      DO 6  I=1,IMAX
      CALL MRFIT(I)
      Y(I+100)=FF
    6 Y(I+200)=(Y(I+100)/Y(I)-1.)*100.
      WRITE(6,4)(PAR(I),Y(I),Y(I+100),Y(I+200),I=1,IMAX)
    4 FORMAT(1H ,4G15.7)
      STOP
      END
```

```
      SUBROUTINE MRFIT(I)
C-----------------------------------------------------------------------
C     CALCULATES FUNCTION FF AND DERIVATIVES A(I) FOR PROGRAM MUROFIT
C-----------------------------------------------------------------------
      COMMON IMAX,Y(300),P(300),NX,NXG,X(32),FF,A(32),NB,NGB,NF,Q,PF,
     *FV(32),FFV(5),NT,PAR(300)
      POWER=-X(1)
      A(2)= PAR(I)**POWER
      FF=A(2)*X(2)
      A(1)= -FF*ALOG(PAR(I))
      RETURN
      END
```